

9. AN INTERACTIVE REAL-WORLD CARD CONTROLLER PROGRAM

The program below can be used to communicate interactively with any of the devices (ports) on any RWC module. It can be used to test any of the RWC modules by reading data from, or by writing data to, any selected device on the module. Any combination of RWC modules may be used with different address settings.

DAI 8080 ASSEMBLY SERVICE, D2. 2

PAGE 1

```

; DATA APPLICATIONS INTERNATIONAL.
; SIMPLE INTERACTIVE REAL-WORLD CARD CONTROLLER.
; V 1. 0
; 19 MAY 78.

; THIS PROGRAM ALLOWS THE USER TO COMMUNICATE INTERACTIVELY
; WITH ANY REAL-WORLD CARD IN A DCE SYSTEM.
; ON ENTRY THE PROGRAM PRINTS:
;     RWC CONTROLLER
;     P=PUT, G=GET, I=INIT (RIC MODE), C=COMMENT
;     01
; THE USER MUST NOW TYPE ANY OF THE ABOVE COMMAND LETTERS.
; 01 IS THE LINE NUMBER (GIVEN TO EACH COMMAND SEQUENCE).
; DEPENDING ON THE INPUT COMMAND, THE PROGRAM REQUESTS THE
; NECESSARY PARAMETERS. FOR EXAMPLE 'CARD', AFTER WHICH THE
; USER MUST TYPE THE HEX ADDRESS OF THE RWC TO BE ADDRESSED.
; AFTER A PUT COMMAND, THE VALUE OF THE DATA MUST BE TYPED
; IN HEX. AFTER A 'GET' COMMAND TO A SPECIFIC RWC CARD PORT,
; THE PROGRAM READS THE DATA AND PRINTS IT.

; COMMAND FORMATS:

; P = "PUT"
; PUT NN CARD X PORT Y.

; G = "GET"
; GET CARD X PORT Y. DATA= NN

; I = "INIT" (RIC MODE CHANGE).
; INIT CARD X A-MODE M B-MODE N

; C = "COMMENT". TEXT... CR.
; LINE OF COMMENT.

; WHERE: -
; NN IS DATA (2 HEX DIGITS)
; X IS THE REAL-WORLD CARD ADDRESS (1-F)
; Y IS THE RIC PORT OR REGISTER ADDRESS
; M IS THE RIC A-MODE
; N IS THE RIC B-MODE
; NOTE: PROGRAM WRITES 'XY' TO DCE GIC PORT 1.

; LINES ARE NUMBERED FOR REFERENCE.

; "ESC" TERMINATES THE LINE WITH NO EXECUTION.

```

DAI 8080 ASSEMBLY SERVICE, D2. 2

PAGE 2

; EQUATES

054C	CI	EQU	054CH
0561	CIE	EQU	0561H
053F	CO	EQU	053FH
053A	TSP	EQU	053AH
061F	TCRLF	EQU	061FH
0602	TBYTE	EQU	0602H
060B	THEX	EQU	060BH
0614	COMSG	EQU	0614H
031E	RDRWC	EQU	031EH
0349	WRRWC	EQU	0349H
01F0	ASHEX	EQU	01F0H
0050	P	EQU	50H
0047	G	EQU	47H
0049	I	EQU	49H
0043	CEE	EQU	43H
0020	SPACE	EQU	20H
001B	ESC	EQU	1BH
003F	?	EQU	3FH
000D	CR	EQU	0DH

0800	ORG	800H	
0800	310011	LXI	SP, 1100H
0803	CD1F06	CALL	TCRLF
0806	217708	LXI	H, HDNG ; "RWC CONTROLLER".
0809	CD1406	CALL	COMSG
080C	CD1F06	CALL	TCRLF
080F	218608	LXI	H, INSTR ; "P=PUT, G=GET, I=INIT".
0812	CD1406	CALL	COMSG
0815	AF	XRA	A
0816	320610	STA	LINO ; SET LINE NO TO 0.
0819	CD1F06	RWCCC: CALL	TCRLF
081C	CDDA09	CALL	NUMBER ; PRINT LINE NO.
081F	CD4C05	RWCCA: CALL	CI ; WAIT FOR P, G, I, C.
0822	FE50	CPI	P
0824	CA3F08	JZ	PUT
0827	FE47	CPI	G
0829	CA5408	JZ	GET
082C	FE49	CPI	I
082E	CAF208	JZ	INIT
0831	FE43	CPI	CEE
0833	CA6608	JZ	COM
0836	C31F08	JMP	RWCCA
0839	310011	RWCCB: LXI	SP, 1100H ; STRAIGHTEN STACK
083C	C31908	JMP	RWCCC ; AFTER "ESC".
083F	21B108	PUT: LXI	H, MPUT ; "PUT "
0842	CD1406	CALL	COMSG
0845	CD4109	PUTA: CALL	RDATA ; ENTER DATA.
0848	CD1C09	CALL	RBOARD ; ENTER CARD ADDR.
084B	CD3409	CALL	RPORT ; ENTER PORT.
084E	CDB009	CALL	RPUT ; SEND DATA TO CARD, PORT
0851	C31908	JMP	RWCCC
0854	21B608	GET: LXI	H, MGET ; "GET FROM "
0857	CD1406	CALL	COMSG
085A	CD1C09	CALL	RCARD ; ENTER CARD ADDR.
085D	CD3409	CALL	RPORT ; ENTER PORT.
0860	CDBA09	CALL	RGET ; GET DATA & PRINT.
0863	C31908	JMP	RWCCC
0866	21E808	COM: LXI	H, MCOM ; "COMMENT- "
0869	CD1406	CALL	COMSG
086C	CD6105	COMA: CALL	CIE
086F	FE0D	CPI	CR ; END OF LINE?.
0871	CA1908	JZ	RWCCC
0874	C36C08	JMP	COMA

```

0877 52574320 HDNG:    DB      'RWC CONTROLLER',0
087B 434F4E54
087F 524F4C4C
0883 455200
0886 503D5055 INSTR:  DB      'P=PUT, G=GET, I=INIT (RIC MODE), C=COMMENT',0
088A 542C2047
088E 3D474554
0892 2C20493D
0896 494E4954
089A 20285249
089E 43204D4F
08A2 4445292C
08A6 20433D43
08AA 4F4D4D45
08AE 4E5400
08B1 50555420 MPUT:   DB      'PUT ',0
08B5 00
08B6 47455420 MGET:   DB      'GET ',0
08BA 00
08BB 494E4954 MINIT:  DB      'INIT ',0
08BF 2000
08C1 43415244 MCARD:  DB      'CARD ',0
08C5 2000
08C7 504F5254 MPORT:  DB      'PORT ',0
08CB 2000
08CD 20412D4D MAMOD:  DB      ' A-MODE ',0
08D1 4F444520
08D5 00
08D6 20422D4D MBMOD:  DB      ' B-MODE ',0
08DA 4F444520
08DE 00
08DF 2E204441 MDATA:  DB      '. DATA= ',0
08E3 54413D20
08E7 00
08E8 434F4D4D MCOM:   DB      'COMMENT- ',0
08EC 454E542D
08F0 2000

```

```

08F2 21BB08  INIT:  LXI      H, MINIT      ; "INIT "
08F5 CD1406          CALL     COMSG
08F8 CD1C09          CALL     RCARD      ; ENTER CARD ADDR.
08FB 3E03           MVI      A, 3
08FD 320110         STA      PORT      ; RIC COMMAND REG.
0900 CD8C09         CALL     RAMOD     ; ENTER A-MODE.
0903 CD9E09         CALL     RBMOD     ; ENTER B-MODE.
0906 3A0210         LDA      AMODE
0909 87             ADD      A
090A 87             ADD      A
090B 87             ADD      A      ; *8.
090C C680           ADI      80H      ; +80
090E 47             MOV      B, A      ; SAVE.
090F 3A0310         LDA      BMODE
0912 80             ADD      B      ; 80H+8*AMODE+BMODE.
0913 320410         STA      DATA0
0916 CDB009         CALL     RPUT      ; SEND.
0919 C31908         JMP      RWCCC

```

```

; RCARD. PRINT "CARD ". ENTER DIGIT (1-F).
; EXIT WITH "SPACE".

```

```

091C 21C108  RCARD:  LXI      H, MCARD      ; "CARD ".
091F CD1406          CALL     COMSG
0922 AF           XRA      A      ; ZERO.
0923 320010         STA      CARD      ; DEFAULT ADDRESS.
0926 CD6409  RCDA:  CALL     RDIGIT
0929 B7           ORA      A
092A CC8409         CZ      WHAT
092D CA2609         JZ      RCDA      ; INVALID.
0930 320010         STA      CARD
0933 C9           RET

```

```

; RPORT. PRINT "PORT ". ENTER PORT (0-3).
; EXIT WITH SPACE.

```

```

0934 21C708  RPORT:  LXI      H, MPORT      ; "PORT ".
0937 CD1406          CALL     COMSG
093A CD6409  RPTA:  CALL     RDIGIT
093D 320110         STA      PORT
0940 C9           RET

```

;RDATA. ENTER HEX BYTE.
;EXIT WITH "SPACE".

```

0941 CD6105  RDATA:  CALL    CIE
0944 FE20      CPI     SPACE
0946 C8        RZ
0947 FE1B      CPI     ESC
0949 CA3908    JZ      RWCCB      ; ABORT LINE.
094C CDF001    CALL    ASHEX
094F DC8409    CC     WHAT
0952 DA4109    JC     RDATA
0955 47        MOV     B, A      ; SAVE.
0956 3A0410    LDA     DATA0    ; GET OLD BYTE.
0959 87        ADD     A
095A 87        ADD     A
095B 87        ADD     A
095C 87        ADD     A      ; SHIFT LEFT 4.
095D 80        ADD     B      ; ADD IN NEW DIGIT.
095E 320410    STA     DATA0
0961 C34109    JMP     RDATA      ; NEXT.

```

;RDIGIT. ENTER HEX DIGIT, EXIT WITH "SPACE"

```

0964 CD6105  RDIGIT: CALL    CIE      ; ENTER/PRINT DIGIT.
0967 FE20      CPI     SPACE
0969 CA8009    JZ      RDTA      ; RETURN IF SPACE.
096C FE1B      CPI     ESC
096E CA3908    JZ      RWCCB    ; ABORT LINE.
0971 CDF001    CALL    ASHEX    ; CONVERT TO HEX.
0974 DC8409    CC     WHAT    ; PRINT "?".
0977 DA6409    JC     RDIGIT
097A 320510    STA     DIGIT
097D C36409    JMP     RDIGIT    ; NEXT.
0980 3A0510    RDTA:  LDA     DIGIT    ; GET (DIGIT) FOR RET.
0983 C9        RET

```

;WHAT. PRINT "?".

```

0984 F5        WHAT:  PUSH   PSW      ; SAVE A-REG, FLAGS.
0985 0E3F      MVI   C, ?
0987 CD3F05    CALL  CO
098A F1        POP    PSW
098B C9        RET

```

```

; RAMOD. PRINT " A-MODE ". ENTER DIGIT (0-8).
098C 21CD08 RAMOD: LXI H, MAMOD ; " A-MODE ".
098F CD1406 CALL COMSG
0992 CD6409 RAMA: CALL RDIGIT
0995 FE09 CPI 9
0997 D29209 JNC RAMA ; INVALID.
099A 320210 STA AMODE
099D C9 RET

; RBMOD. PRINT " B-MODE ". ENTER DIGIT (0-6).
099E 21D608 RBMOD: LXI H, MBMOD ; " B-MODE ".
09A1 CD1406 CALL COMSG
09A4 CD6409 RBMA: CALL RDIGIT
09A7 FE07 CPI 7
09A9 D2A409 JNC RBMA
09AC 320310 STA BMODE
09AF C9 RET

; RPUT. SEND DATA0 TO CARD, PORT.
09B0 CDCA09 RPUT: CALL RWCAD ; ADDRESS RWC.
09B3 3A0410 LDA DATA0 ; GET DATA TO SEND.
09B6 CD4903 CALL WRRWC ; REAL-WORLD WRITE.
09B9 C9 RET

; RGET. READ CARD, PORT & PRINT DATA.
09BA 21DF08 RGET: LXI H, MDATA ; ". DATA = ".
09BD CD1406 CALL COMSG
09C0 CDCA09 CALL RWCAD
09C3 CD1E03 CALL RDRWC ; READ RWC.
09C6 CD0206 CALL TBYTE
09C9 C9 RET

```

; RWCAD. ADDRESS RWC. (CARD)*8+(PORT) > GIC PORT1.

```

09CA 3A0010  RWCAD:  LDA    CARD
09CD 87      ADD    A
09CE 87      ADD    A
09CF 87      ADD    A
09D0 87      ADD    A
09D1 47      MOV    B, A
09D2 3A0110  LDA    PORT
09D5 80      ADD    B
+          STGI   1
09D6 32011C
09D9 C9      RET

```

; NUMBER. UPDATE & PRINT LINE NO.

```

09DA 3A0610  NUMBER:  LDA    LINO
09DD 3C      INR    A
09DE FE64    CPI    100      ; 0-99 ONLY.
09E0 C2E409  JNZ    NRA
09E3 AF      XRA    A          ; RESET.
09E4 320610  NRA:    STA    LINO      ; UPDATE.
09E7 0E2F    MVI    C, 2FH      ; TENS (-1+30H FOR ASCII "-1")
09E9 0C      NRB:    INR    C          ; COUNT TENS.
09EA D60A    SUI    10          ; SUBTRACT 10 TIL -VE.
09EC F2E909  JP     NRB
09EF C63A    ADI    3AH
09F1 47      MOV    B, A          ; STRAIGHTEN UNITS & MAKE ASCI
09F2 CD3F05  CALL   CO            ; SAVE.
09F5 48      MOV    C, B        ; PRINT TENS.
                                ; GET UNITS
                                ; PRINT
09F9 CD3A05  CALL   TSP
09FC C9      RET
1000      ORG    1000H

1000 00      CARD:   DB    0
1001 00      PORT:   DB    0
1002 00      AMODE:  DB    0
I 1003 00      BMODE:  DB    0
1004 00      DATA0: DB    0
1005 00      DIGIT:  DB    0
1006 00      LINO:   DB    0

```

0000 END

?	003F	AMODE	1002	ASHEX	01F0	BASE	0000
BCLR	4FA0	BMODE	1003	BSET	4FBD	CARD	1000
CEE	0043	CI	054C	CIE	0561	CO	053F
COM	0866	COMA	086C	COMSG	0614	CR	000D
DATA0	1004	DCE1	FFFF	DCE2	0000	DCEX	0000
DCEXB	0000	DIGIT	1005	ENDRM	1200	ESC	001B
FALSE	0000	G	0047	GET	0854	GIC	1C00
GICC	4FDC	GICS	5C00	HDNG	0877	I	0049
INIT	08F2	INSTR	0886	JEQ	4E3A	JGE	4E18
JGT	4EOC	JLE	4E01	JLT	4E23	JNE	4E2E
LDGI	4F90	LDGIS	4F71	LDIN	4F20	LDIRP	4EB8
LDRCV	4F42	LDSTA	4EEB	LINO	1006	MAMOD	08CD
MBMOD	08D6	MCARD	08C1	MCOM	08E8	MDATA	08DF
MGET	08B6	MINIT	08BB	MPORT	08C7	MPUT	08B1
NRA	09E4	NRB	09E9	NUMBE	09DA	ORGRM	1000
P	0050	PAGE	4EB1	PORT	1001	PUT	083F
PUTA	0845	RAMA	0992	RAMOD	098C	RBMA	09A4
RBMOD	099E	RCARD	091C	RCDA	0926	RDATA	0941
RDIGI	0964	RDRWC	031E	RDTA	0980	RGET	09BA
RPORT	0934	RPTA	093A	RPUT	09B0	RWBCL	4E45
RWBST	4E63	RWCAD	09CA	RWCCA	081F	RWCCB	0839
RWCCC	0819	RWGIC	4E83	SPACE	0020	STCRR	4EC9
STGI	4F80	STGIS	4F62	STIMR	4EFC	STOUT	4F31
STTCM	4EDA	STTIM	4F0D	STXMT	4F51	TBYTE	0602
TCRLF	061F	THEX	060B	TICC	9800	TRUE	FFFF
TSP	053A	WHAT	0984	WRRWC	0349		

DCE UTILITY V2. 0

>Z3

>G800

RWC CONTROLLER

P=PUT, G=GET, I=INIT (RIC MODE), C=COMMENT

01 COMMENT- THIS IS AN EXAMPLE RUN

02 COMMENT- SET RWC-T24 WITH ADDR=7 FOR ALL PORTS 0/P

03 INIT CARD 7 A-MODE 0 B-MODE 0

04 COMMENT- NOW SEND 55H TO PORT 0

05 PUT 55 CARD 7 PORT 0

06 COMMENT- AND 73H TO PORT 2

07 PUT 73 CARD 7 PORT 2

08 COMMENT- CHANGE RIC MODE TO I/P

09 INIT CARD 7 A-MODE 3 B-MODE 3

10 COMMENT- AND READ PORT 1

11 GET CARD 7 PORT 1 . DATA=FF

12 COMMENT- FF WAS READ FROM PORT 1

13 COMMENT- END OF RUN.

14